Non-Homogeneous Second-Order Differential Equations

James Rohal

We are going to use Maple to solve some second-order differential equations. They take the form $P(x)\frac{d^2y}{dx^2} + Q(x)\frac{dy}{dx} + R(x)y = G(x)$. A non-homogeneous second-order differential equations has $G(x) \neq 0$. We are going to look at a special case where $P(x) = a$, $Q(x) = b$, $R(x) = c$ and a, b, c are constants. The given by $y(x) = y_c(x) + y_p(x)$ where $y_c(x)$ is the complementary solution and $y_p(x)$ is the particular solution. The example we will be considering is $y'' + y' - 6y = G(x)$. In an earlier worksheet we found that $y_c(x) = c_1 e^{2x} + c_2 e^{-3x}$ by studying the characteristic polynomial. Below we show how to find $y_p(x)$ for different $G(x)$.

$G(x) = polynomial$

If *polynomial* has degree n, then we guess that $y_p(x) = a_n x^n + a_{n-1}x^{n-1} + \ldots + a_1 x^1 + a_0 x^0$ where $a_n, a_{n-1}, \ldots, a_1, a_0$ are constants.

Suppose $G(x) = x^2$. The degree is 2 so we guess $y_p(x) = A x^2 + B x + C$. (It doesn't matter what we call the constants.) Now we use the method of undetermined coefficients. First, we find the derivatives of the particular solution.

```
> yp              := A*x^2 + B*x + C;
  yp_prime        := diff(yp, x);
  yp_double_prime := diff(yp, x$2);
```

$$yp := A x^2 + B x + C$$
$$yp\_prime := 2 A x + B$$
$$yp\_double\_prime := 2 A \quad \textbf{(1.1)}$$

Now we substitute the expressions we found above in to the differential equation.

```
> ode := diff(y(x), x$2) + diff(y(x), x) - 6*y(x) = x^2;

  ode_with_yp := eval(ode, {diff(y(x), x$2) = yp_double_prime,
  diff(y(x), x) = yp_prime, y(x) = yp});
```

$$ode := \frac{d^2}{dx^2} y(x) + \frac{d}{dx} y(x) - 6 y(x) = x^2$$
$$ode\_with\_yp := 2 A + 2 A x + B - 6 A x^2 - 6 B x - 6 C = x^2 \quad \textbf{(1.2)}$$

We then collect common terms.

```
> collect(ode_with_yp, x);
```

$$-6 A x^2 + (-6 B + 2 A) x + 2 A - 6 C + B = x^2 \quad \textbf{(1.3)}$$

If the particular solution is in fact a solution to our ODE, then the coefficients on the LHS must agree with the coefficients on the RHS. This gives us a system of equations to solve.

```
> constants :=
  solve([
         (-6)*A + ( 0)*B + ( 0)*C = 1,
         ( 2)*A + (-6)*B + ( 0)*C = 0,
         ( 2)*A + ( 1)*B + (-6)*C = 0
```

```
   ],
  {A,B,C});
```

$$constants := \left\{A = -\frac{1}{6}, B = -\frac{1}{18}, C = -\frac{7}{108}\right\} \qquad \textbf{(1.4)}$$

Thus the particular solution is

```
> eval(yp, constants);
```

$$-\frac{1}{6}\,x^2 - \frac{1}{18}\,x - \frac{7}{108} \qquad \textbf{(1.5)}$$

The general solution is then $y(x) = y_c(x) + y_p(x) = \left(c_1 e^{2x} + c_2 e^{-3x}\right) + \left(-\frac{7}{108} - \frac{1}{18}\,x - \frac{1}{6}\,x^2\right)$.

We can verify this using **dsolve**.

```
> dsolve(diff(y(x), x$2) + diff(y(x), x) - 6*y(x) = x^2, y(x));
```

$$y(x) = \_C2\, e^{-3x} + \_C1\, e^{2x} - \frac{7}{108} - \frac{1}{18}\,x - \frac{1}{6}\,x^2 \qquad \textbf{(1.6)}$$

$G(x) = e^{k \cdot x}$

We guess that $y_p(x) = A\,e^{k \cdot x}$ where A is a constant.

Suppose $G(x) = e^{3x}$. Here $k = 3$, so we guess $y_p(x) = A\,e^{3x}$. Now we use the method of undetermined coefficients. First, we find the derivatives of the particular solution.

```
> yp              := A*exp(3*x);
  yp_prime        := diff(yp, x);
  yp_double_prime := diff(yp, x$2);
```

$$yp := A\,e^{3x}$$

$$yp\_prime := 3\,A\,e^{3x}$$

$$yp\_double\_prime := 9\,A\,e^{3x} \qquad \textbf{(2.1)}$$

Now we substitute the expressions we found above in to the differential equation.

```
> ode := diff(y(x), x$2) + diff(y(x), x) - 6*y(x) = exp(3*x);

  ode_with_yp := eval(ode, {diff(y(x), x$2) = yp_double_prime,
  diff(y(x), x) = yp_prime, y(x) = yp});
```

$$ode := \frac{d^2}{dx^2}\,y(x) + \frac{d}{dx}\,y(x) - 6\,y(x) = e^{3x}$$

$$ode\_with\_yp := 6\,A\,e^{3x} = e^{3x} \qquad \textbf{(2.2)}$$

If the particular solution is in fact a solution to our ODE, then the coefficients on the LHS must agree with the coefficients on the RHS. This says that $6\,A = 1 \Rightarrow A = \frac{1}{6}$. Thus the particular solution is $y_p(x) = \frac{1}{6}\,e^{3x}$ and the general solution is $y(x) = y_c(x) + y_p(x) = \left(c_1 e^{2x} + c_2 e^{-3x}\right) + \left(\frac{1}{6}\,e^{3x}\right)$.

We can verify this using **dsolve**.

```
> dsolve(diff(y(x), x$2) + diff(y(x), x) - 6*y(x) = exp(3*x), y
  (x));
```

(2.3)

$$y(x) = e^{2x}\_C2 + e^{-3x}\_C1 + \frac{1}{6} e^{3x} \quad \textbf{(2.3)}$$

$G(x) = \cos(m \cdot x)$ **or** $\sin(m \cdot x)$

We guess that $y_p(x) = A\cos(m\,x) + B\sin(m\,x)$ where A, B are constants.

Suppose $G(x) = \cos(3\,x)$. Here $m = 3$ so we guess $y_p(x) = A\cos(3\,x) + B\sin(3\,x)$. Now we use the method of undetermined coefficients. First, we find the derivatives of the particular solution.

```
> yp              := A*cos(3*x) + B*sin(3*x);
  yp_prime        := diff(yp, x);
  yp_double_prime := diff(yp, x$2);
```

$$yp := A\cos(3\,x) + B\sin(3\,x)$$

$$yp\_prime := -3\,A\sin(3\,x) + 3\,B\cos(3\,x)$$

$$yp\_double\_prime := -9\,A\cos(3\,x) - 9\,B\sin(3\,x) \quad \textbf{(3.1)}$$

Now we substitute the expressions we found above in to the differential equation.

```
> ode := diff(y(x), x$2) + diff(y(x), x) - 6*y(x) = cos(3*x);

  ode_with_yp := eval(ode, {diff(y(x), x$2) = yp_double_prime,
  diff(y(x), x) = yp_prime, y(x) = yp});
```

$$ode := \frac{d^2}{dx^2}\,y(x) + \frac{d}{dx}\,y(x) - 6\,y(x) = \cos(3\,x)$$

$$ode\_with\_yp := -15\,A\cos(3\,x) - 15\,B\sin(3\,x) - 3\,A\sin(3\,x) + 3\,B\cos(3\,x) = \cos(3\,x) \quad \textbf{(3.2)}$$

We then collect common terms. We have to collect the terms with $\cos(3\,x)$ and $\sin(3\,x)$ separately.

```
> collected_ode := collect(ode_with_yp, cos(3*x));
  collected_ode := collect(collected_ode, sin(3*x));
```

$$collected\_ode := (-15\,A + 3\,B)\cos(3\,x) - 15\,B\sin(3\,x) - 3\,A\sin(3\,x) = \cos(3\,x)$$

$$collected\_ode := (-15\,B - 3\,A)\sin(3\,x) + (-15\,A + 3\,B)\cos(3\,x) = \cos(3\,x) \quad \textbf{(3.3)}$$

If the particular solution is in fact a solution to our ODE, then the coefficients on the LHS must agree with the coefficients on the RHS. This gives us a system of equations to solve.

```
> constants :=
  solve([
         ( -3)*A + (-15)*B = 0,
         (-15)*A + (  3)*B = 1
        ],
  {A,B});
```

$$constants := \left\{A = -\frac{5}{78}, B = \frac{1}{78}\right\} \quad \textbf{(3.4)}$$

Thus the particular solution is

```
> eval(yp, constants);
```

$$-\frac{5}{78}\cos(3\,x) + \frac{1}{78}\sin(3\,x) \quad \textbf{(3.5)}$$

The general solution is then

$y(x) = y_c(x) + y_p(x) = \left(c_1 e^{2x} + c_2 e^{-3x}\right) + \left(-\frac{5}{78}\cos(3\,x) + \frac{1}{78}\sin(3\,x)\right).$

We can verify this using `dsolve`.

```
> dsolve(diff(y(x), x$2) + diff(y(x), x) - 6*y(x) = cos(3*x), y
  (x));
```

$$y(x) = \_C2\, e^{-3x} + \_C1\, e^{2x} - \frac{5}{78}\cos(3x) + \frac{1}{78}\sin(3x) \quad \textbf{(3.6)}$$

$G(x) = e^{kx} P(x)\cos(mx)$ or $e^{kx} P(x)\sin(mx)$

This is the most general case, as it is a product of cases above. If $P(x)$ is a n^{th} degree polynomial then our guess should be $y_p(x) = e^{kx}Q(x)\cos(mx) + e^{kx}R(x)\sin(mx)$.

Suppose $G(x) = e^{2x}\cdot x\cdot\cos(3x)$. Here $k=2$, $P(x) = x$, $m=3$ so we guess $y_p(x) = e^{2x}(A\cdot x + B)\cos(3x) + e^{2x}(C\cdot x + F)\sin(3x)$. (I choose to use F because D is reserved by Maple). Now we use the method of undetermined coefficients. First, we find the derivatives of the particular solution.

```
> yp              := exp(2*x)*(A*x+B)*cos(3*x) + exp(2*x)*(C*x+F)
  *sin(3*x);
  yp_prime        := diff(yp, x);
  yp_double_prime := diff(yp, x$2);
```

$$yp := e^{2x}(Ax+B)\cos(3x) + e^{2x}(Cx+F)\sin(3x)$$

$$yp\_prime := 2e^{2x}(Ax+B)\cos(3x) + e^{2x}A\cos(3x) - 3e^{2x}(Ax+B)\sin(3x) + 2e^{2x}(Cx+F)\sin(3x) + e^{2x}C\sin(3x) + 3e^{2x}(Cx+F)\cos(3x)$$

$$yp\_double\_prime := -5e^{2x}(Ax+B)\cos(3x) + 4e^{2x}A\cos(3x) - 12e^{2x}(Ax+B)\sin(3x) - 6e^{2x}A\sin(3x) - 5e^{2x}(Cx+F)\sin(3x) + 4e^{2x}C\sin(3x) + 12e^{2x}(Cx+F)\cos(3x) + 6e^{2x}C\cos(3x) \quad \textbf{(4.1)}$$

Now we substitute the expressions we found above in to the differential equation.

```
> ode := diff(y(x), x$2) + diff(y(x), x) - 6*y(x) = exp(2*x)*x*
  cos(3*x);

  ode_with_yp := eval(ode, {diff(y(x), x$2) = yp_double_prime,
  diff(y(x), x) = yp_prime, y(x) = yp});
```

$$ode := \frac{d^2}{dx^2}y(x) + \frac{d}{dx}y(x) - 6y(x) = e^{2x}x\cos(3x)$$

$$ode\_with\_yp := -9e^{2x}(Ax+B)\cos(3x) + 5e^{2x}A\cos(3x) - 15e^{2x}(Ax+B)\sin(3x) - 6e^{2x}A\sin(3x) - 9e^{2x}(Cx+F)\sin(3x) + 5e^{2x}C\sin(3x) + 15e^{2x}(Cx+F)\cos(3x) + 6e^{2x}C\cos(3x) = e^{2x}x\cos(3x) \quad \textbf{(4.2)}$$

We then collect common terms. We have to collect the terms with $e^{2x}x\cos(3x)$, $e^{2x}x\sin(3x)$, $e^{2x}\cos(3x)$, $e^{2x}\sin(3x)$ separately. This isn't the easiest thing in Maple so we do the best we can.

```
> collected_ode := collect(ode_with_yp, sin(3*x));
  collected_ode := collect(collected_ode, cos(3*x));
  collected_ode := collect(collected_ode, x);
  collected_ode := collect(collected_ode, exp(2*x));
```

$$collected\_ode := \left(-9e^{2x}(Cx+F) + 5e^{2x}C - 15e^{2x}(Ax+B) - 6e^{2x}A\right)\sin(3x) - 9e^{2x}(Ax+B)\cos(3x) + 5e^{2x}A\cos(3x) + 15e^{2x}(Cx+F)\cos(3x)$$

$$+6e^{2x}C\cos(3x) = e^{2x}x\cos(3x)$$

$$collected\_ode := \left(-9e^{2x}(Ax+B) + 5e^{2x}A + 15e^{2x}(Cx+F) + 6e^{2x}C\right)\cos(3x) + \left(-9e^{2x}(Cx+F) + 5e^{2x}C - 15e^{2x}(Ax+B) - 6e^{2x}A\right)\sin(3x) = e^{2x}x\cos(3x)$$

$$collected\_ode := \left(\left(15e^{2x}C - 9e^{2x}A\right)\cos(3x) + \left(-15e^{2x}A - 9e^{2x}C\right)\sin(3x)\right)x + \left(-9e^{2x}B + 6e^{2x}C + 5e^{2x}A + 15e^{2x}F\right)\cos(3x) + \left(-9e^{2x}F - 6e^{2x}A + 5e^{2x}C - 15e^{2x}B\right)\sin(3x) = e^{2x}x\cos(3x)$$

$$collected\_ode := \left(\left((15C-9A)\cos(3x) + (-15A-9C)\sin(3x)\right)x + (-9B+6C+5A+15F)\cos(3x) + (-9F-6A+5C-15B)\sin(3x)\right)e^{2x} = e^{2x}x\cos(3x) \quad \textbf{(4.3)}$$

If the particular solution is in fact a solution to our ODE, then the coefficients on the LHS must agree with the coefficients on the RHS. This gives us a system of equations to solve.

```
> constants :=
  solve([
          ( -9)*A + (  0)*B + (15)*C + ( 0)*F = 1,
          (-15)*A + (  0)*B + (-9)*C + ( 0)*F = 0,
          (  5)*A + ( -9)*B + ( 6)*C + (15)*F = 0,
          ( -6)*A + (-15)*B + ( 5)*C + (-9)*F = 0
         ],
  {A,B,C,F});
```

$$constants := \left\{A = -\frac{1}{34}, B = \frac{65}{2601}, C = \frac{5}{102}, F = \frac{3}{578}\right\} \quad \textbf{(4.4)}$$

Thus the particular solution is

```
> eval(yp, constants);
```

$$e^{2x}\left(-\frac{1}{34}x + \frac{65}{2601}\right)\cos(3x) + e^{2x}\left(\frac{5}{102}x + \frac{3}{578}\right)\sin(3x) \quad \textbf{(4.5)}$$

The general solution is then

$$y(x) = y_c(x) + y_p(x) = \left(c_1e^{2x} + c_2e^{-3x}\right) + \left(e^{2x}\left(-\frac{1}{34}x + \frac{65}{2601}\right)\cos(3x) + e^{2x}\left(\frac{5}{102}x + \frac{3}{578}\right)\sin(3x)\right)$$

.

We can verify this using `dsolve`. It turns out these solutions are the same.

```
> dsolve(diff(y(x), x$2) + diff(y(x), x) - 6*y(x) = exp(2*x)*x*
  cos(3*x), y(x));
```

$$y(x) = \_C2\, e^{-3x} + \_C1\, e^{2x} + \frac{1}{5202}(-153x + 130)e^{2x}\cos(3x) + \frac{5}{102}\sin(3x)e^{2x}\left(x + \frac{9}{85}\right) \quad \textbf{(4.6)}$$

Homework

Solve the differential equation $y'' - y = xe^x$, $y(0) = 2$, $y'(0) = 1$.